NAG Library Routine Document

E02CBF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

E02CBF evaluates a bivariate polynomial from the rectangular array of coefficients in its double Chebyshev series representation.

2 Specification

```
SUBROUTINE E02CBF (MFIRST, MLAST, K, L, X, XMIN, XMAX, Y, YMIN, YMAX,      &
                   FF, A, NA, WORK, NWORK, IFAIL)
INTEGER            MFIRST, MLAST, K, L, NA, NWORK, IFAIL
REAL (KIND=nag_wp) X(MLAST), XMIN, XMAX, Y, YMIN, YMAX, FF(MLAST),         &
                   A(NA), WORK(NWORK)
```

3 Description

This subroutine evaluates a bivariate polynomial (represented in double Chebyshev form) of degree k in one variable, $\bar{x}$, and degree l in the other, $\bar{y}$. The range of both variables is -1 to $+1$. However, these normalized variables will usually have been derived (as when the polynomial has been computed by E02CAF, for example) from your original variables x and y by the transformations

$$\bar{x} = \frac{2x - (x_{\text{max}} + x_{\text{min}})}{(x_{\text{max}} - x_{\text{min}})} \quad \text{and} \quad \bar{y} = \frac{2y - (y_{\text{max}} + y_{\text{min}})}{(y_{\text{max}} - y_{\text{min}})}.$$

(Here x_{min} and x_{max} are the ends of the range of x which has been transformed to the range -1 to $+1$ of $\bar{x}$. y_{min} and y_{max} are correspondingly for y. See Section 9). For this reason, the subroutine has been designed to accept values of x and y rather than $\bar{x}$ and $\bar{y}$, and so requires values of x_{min}, etc. to be supplied by you. In fact, for the sake of efficiency in appropriate cases, the routine evaluates the polynomial for a sequence of values of x, all associated with the same value of y.

The double Chebyshev series can be written as

$$\sum_{i=0}^{k} \sum_{j=0}^{l} a_{ij} T_i(\bar{x}) T_j(\bar{y}),$$

where $T_i(\bar{x})$ is the Chebyshev polynomial of the first kind of degree i and argument $\bar{x}$, and $T_j(\bar{y})$ is similarly defined. However the standard convention, followed in this subroutine, is that coefficients in the above expression which have either i or j zero are written $\frac{1}{2}a_{ij}$, instead of simply a_{ij}, and the coefficient with both i and j zero is written $\frac{1}{4}a_{0,0}$.

The subroutine first forms $c_i = \sum_{j=0}^{l} a_{ij} T_j(\bar{y})$, with $a_{i,0}$ replaced by $\frac{1}{2}a_{i,0}$, for each of $i = 0, 1, \ldots, k$. The value of the double series is then obtained for each value of x, by summing $c_i \times T_i(\bar{x})$, with c_0 replaced by $\frac{1}{2}c_0$, over $i = 0, 1, \ldots, k$. The Clenshaw three term recurrence (see Clenshaw (1955)) with modifications due to Reinsch and Gentleman (1969) is used to form the sums.

4 References

Clenshaw C W (1955) A note on the summation of Chebyshev series *Math. Tables Aids Comput.* **9** 118–120

Gentleman W M (1969) An error analysis of Goertzel's (Watt's) method for computing Fourier coefficients *Comput. J.* **12** 160–165

5 Arguments

1: MFIRST – INTEGER *Input*

2: MLAST – INTEGER *Input*

On entry: the index of the first and last x value in the array x at which the evaluation is required respectively (see Section 9).

Constraint: MLAST $\geq$ MFIRST.

3: K – INTEGER *Input*

4: L – INTEGER *Input*

On entry: the degree k of x and l of y, respectively, in the polynomial.

Constraint: K ≥ 0 and L ≥ 0.

5: X(MLAST) – REAL (KIND=nag_wp) array *Input*

On entry: X(i), for $i =$ MFIRST, $\ldots$, MLAST, must contain the x values at which the evaluation is required.

Constraint: XMIN $\leq$ X(i) $\leq$ XMAX, for all i.

6: XMIN – REAL (KIND=nag_wp) *Input*

7: XMAX – REAL (KIND=nag_wp) *Input*

On entry: the lower and upper ends, x_{min} and x_{max}, of the range of the variable x (see Section 3).

The values of XMIN and XMAX may depend on the value of y (e.g., when the polynomial has been derived using E02CAF).

Constraint: XMAX $>$ XMIN.

8: Y – REAL (KIND=nag_wp) *Input*

On entry: the value of the y coordinate of all the points at which the evaluation is required.

Constraint: YMIN $\leq$ Y $\leq$ YMAX.

9: YMIN – REAL (KIND=nag_wp) *Input*

10: YMAX – REAL (KIND=nag_wp) *Input*

On entry: the lower and upper ends, y_{min} and y_{max}, of the range of the variable y (see Section 3).

Constraint: YMAX $>$ YMIN.

11: FF(MLAST) – REAL (KIND=nag_wp) array *Output*

On exit: FF(i) gives the value of the polynomial at the point (x_i, y), for $i =$ MFIRST, $\ldots$, MLAST.

12: A(NA) – REAL (KIND=nag_wp) array *Input*

On entry: the Chebyshev coefficients of the polynomial. The coefficient a_{ij} defined according to the standard convention (see Section 3) must be in A($i \times (l+1) + j + 1$).

13: NA – INTEGER *Input*

On entry: the dimension of the array A as declared in the (sub)program from which E02CBF is called.

Constraint: $\text{NA} \geq (\text{K}+1) \times (\text{L}+1)$, the number of coefficients in a polynomial of the specified degree.

14: WORK(NWORK) – REAL (KIND=nag_wp) array *Workspace*

15: NWORK – INTEGER *Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which E02CBF is called.

Constraint: $\text{NWORK} \geq \text{K}+1$.

16: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: $\text{IFAIL} = 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry $\text{IFAIL} = 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

$\text{IFAIL} = 1$

On entry, $\text{MFIRST} > \text{MLAST}$,
or $\text{K} < 0$,
or $\text{L} < 0$,
or $\text{NA} < (\text{K}+1) \times (\text{L}+1)$,
or $\text{NWORK} < \text{K}+1$.

$\text{IFAIL} = 2$

On entry, $\text{YMIN} \geq \text{YMAX}$,
or $\text{Y} < \text{YMIN}$,
or $\text{Y} > \text{YMAX}$.

$\text{IFAIL} = 3$

On entry, $\text{XMIN} \geq \text{XMAX}$,
or $\text{X}(i) < \text{XMIN}$, or $\text{X}(i) > \text{XMAX}$, for some $i = \text{MFIRST}, \text{MFIRST}+1, \ldots, \text{MLAST}$.

$\text{IFAIL} = -99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -999$

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

The method is numerically stable in the sense that the computed values of the polynomial are exact for a set of coefficients which differ from those supplied by only a modest multiple of ***machine precision***.

8 Parallelism and Performance

E02CBF is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The time taken is approximately proportional to $(k+1) \times (m+l+1)$, where $m = \text{MLAST} - \text{MFIRST} + 1$, the number of points at which the evaluation is required.

This subroutine is suitable for evaluating the polynomial surface fits produced by the subroutine E02CAF, which provides the real array A in the required form. For this use, the values of $y_{\min}$ and $y_{\max}$ supplied to the present subroutine must be the same as those supplied to E02CAF. The same applies to $x_{\min}$ and $x_{\max}$ if they are independent of y. If they vary with y, their values must be consistent with those supplied to E02CAF (see Section 9 in E02CAF).

The arguments MFIRST and MLAST are intended to permit the selection of a segment of the array X which is to be associated with a particular value of y, when, for example, other segments of X are associated with other values of y. Such a case arises when, after using E02CAF to fit a set of data, you wish to evaluate the resulting polynomial at all the data values. In this case, if the arguments X, Y, MFIRST and MLAST of the present routine are set respectively (in terms of arguments of E02CAF) to X, $\text{Y}(S)$, $1 + \sum_{i=1}^{s-1} \text{M}(i)$ and $\sum_{i=1}^{s} \text{M}(i)$, the routine will compute values of the polynomial surface at all data points which have $\text{Y}(S)$ as their y coordinate (from which values the residuals of the fit may be derived).

10 Example

This example reads data in the following order, using the notation of the argument list above:

N K L	
$\text{A}(i)$,	for $i = 1, 2, \ldots, (\text{K}+1) \times (\text{L}+1)$
YMIN YMAX	
$\text{Y}(i)$ $M(i)$ $\text{XMIN}(i)$ $\text{XMAX}(i)$ $\text{X1}(i)$ $\text{XM}(i)$,	for $i = 1, 2, \ldots, N$.

For each line $\text{Y} = \text{Y}(i)$ the polynomial is evaluated at $M(i)$ equispaced points between $\text{X1}(i)$ and $\text{XM}(i)$ inclusive.

10.1 Program Text

```
    Program e02cbfe

!     E02CBF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
      Use nag_library, Only: e02cbf, nag_wp
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: xmax, xmin, y, ymax, ymin
      Integer                          :: i, ifail, j, k, l, m, m1, m2,        &
                                          mfirst, mlast, na, nwork
      Logical                          :: plot
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: a(:), ff(:), work(:), x(:)
!     .. Intrinsic Procedures ..
      Intrinsic                        :: min, real
!     .. Executable Statements ..
!     Skip heading in data file
      Read (nin,*)

      Read (nin,*) plot

      If (.Not. plot) Then
        Write (nout,*) 'E02CBF Example Program Results'
      End If

      Read (nin,*) k, l, m

      If (plot) Then
        m1 = 1
        m2 = m
        mlast = m
      Else
        m1 = (2*m+3)/7
        m2 = (6*m+3)/7 + 1
        mlast = min(5,m)
      End If

      na = (k+1)*(l+1)
      nwork = k + 1
      Allocate (x(mlast),ff(mlast),a(na),work(nwork))

      Read (nin,*) a(1:na)
      Read (nin,*) ymin, ymax, xmin, xmax

      Do j = 1, mlast
        x(j) = xmin + (xmax-xmin)*real(j-1,kind=nag_wp)/real(mlast-1,kind=     &
          nag_wp)
      End Do
      mfirst = 1

      Do i = m1, m2, m1
        y = ymin + ((ymax-ymin)*real(i-1,kind=nag_wp))/real(m-1,kind=nag_wp)

        ifail = 0
        Call e02cbf(mfirst,mlast,k,l,x,xmin,xmax,y,ymin,ymax,ff,a,na,work,     &
          nwork,ifail)

        If (plot) Then

          Do j = 1, mlast
            Write (nout,99998) y, x(j), ff(j)
          End Do
```

```
          Write (nout,*)
        Else
          Write (nout,*)
          Write (nout,99999) 'Y = ', y
          Write (nout,*)
          Write (nout,*) '  I     X(I)      Poly(X(I),Y)'

          Do j = 1, mlast
            Write (nout,99997) j, x(j), ff(j)
          End Do

        End If

      End Do

99999 Format (1X,A,E13.4)
99998 Format (1X,1P,2E13.4,1P,2E13.4)
99997 Format (1X,I3,1P,2E13.4)
    End Program e02cbfe
```

10.2 Program Data

```
E02CBF Example Program Data
.FALSE.           : output data for plotting
   3   2   20    : k, l, m=no of output points.
15.34820
 5.15073
-2.20140
 1.14719
-0.64419
 0.30464
-0.49010
-0.00314
-6.69912
 0.00153
 3.00033
-0.00022         : Chebyshev coefficients
 0.0 4.0 0.1 4.5 : ymin ymax xmin xmax
```

10.3 Program Results

```
 E02CBF Example Program Results

 Y =    0.1053E+01

   I     X(I)      Poly(X(I),Y)
   1   1.0000E-01   7.3827E+00
   2   1.2000E+00  -2.7648E-01
   3   2.3000E+00  -2.2541E-01
   4   3.4000E+00   3.2750E+00
   5   4.5000E+00   5.9637E+00

 Y =    0.2316E+01

   I     X(I)      Poly(X(I),Y)
   1   1.0000E-01   1.0752E+01
   2   1.2000E+00   2.6132E+00
   3   2.3000E+00  -8.3004E-01
   4   3.4000E+00   1.8462E+00
   5   4.5000E+00   1.2066E+01

 Y =    0.3579E+01

   I     X(I)      Poly(X(I),Y)
   1   1.0000E-01   1.1902E+00
   2   1.2000E+00   8.8478E+00
   3   2.3000E+00   7.4980E+00
   4   3.4000E+00   4.2491E+00
   5   4.5000E+00   6.2093E+00
```

Example Program
Evaluation of Least-squares Bi-variate Polynomial Fit